

```

'Calculate the gcd using an exhaustive search algorithm
|
Private Sub cmdFindGCD_Click()

    Dim Num1 As String, Num2 As String
    Dim A As Long, B As Long, GCD As Long, I As Long, Smaller As Long

    'Store string copies of the values entered by the user
    Num1 = CStr(Val(txtA.Text))
    Num2 = CStr(Val(txtB.Text))

    'Convert the values entered by the user to long integers.
    'Use "Abs" to convert to positive values in case the user enters negative values.
    A = Abs(Val(Num1))
    B = Abs(Val(Num2))

    If A > 0 And B > 0 Then 'Calculate the gcd of two positive integers

        'Find the smaller of A and B
        If A < B Then
            Smaller = A
        Else
            Smaller = B
        End If

        'Start the search at the smaller of A and B
        I = Smaller

        'Search until a common divisor is found. Since the search begins from the
        'largest possible value and works its way down, the first common divisor
        'must be the greatest common divisor.
        Do Until Num1 Mod I = 0 And Num2 Mod I = 0 Or I = 1

            I = I - 1

        Loop

        GCD = I

    ElseIf A = 0 Then 'gcd(0,B)=B
        GCD = B

    ElseIf B = 0 Then 'gcd(A,0)=A
        GCD = A

    Else 'gcd(0,0)=0
        GCD = 0

    End If

    'Display the result
    lblEuclidGCD.Caption = "gcd(" & CStr(Num1) & ", " & CStr(Num2) & ") = " & CStr(GCD)

    'Note that gcd(0,0)=0 by definition. This is done for mathematical convenience.

End Sub

```

```

'Calculate the gcd using the Euclidean algorithm.
Private Sub cmdGcd_Click()

    Dim Num1 As String, Num2 As String
    Dim A As Long, B As Long, Temp As Long

    'Store string copies of the values entered by the user
    Num1 = CStr(Val(txtA.Text))
    Num2 = CStr(Val(txtB.Text))

    'Convert the values entered by the user to long integers.
    'Use "Abs" to convert to positive values in case the user enters negative values.
    A = Abs(Val(Num1))
    B = Abs(Val(Num2))

    'Apply the Euclidean algorithm
    Do Until B = 0

        Temp = B 'Copy the value of "B" to "Temp"
        B = A Mod B 'Set the value of "B" to the remainder of "A" divided by "B"
        A = Temp 'Set the value of "A" to the original value of "B"

    Loop

    'Display the result
    lblGcd.Caption = "gcd(" & Num1 & ", " & Num2 & ") = " & CStr(A)

    'Note that gcd(0,0)=0 by definition. This is done for mathematical convenience.

End Sub

```